
11.07 Analytic geometry with 2DGeometry

This section describes the part of this book�s 2DGeometry package that manipulates lines
as in conventionalplane analytic geometry.  It�s used to compute and reason with the
coordinates of the points and equations of the lines that constitute a finished drawing.

The  lineq  ADT

The 2DGeometry package uses objects of an abstract data type  lineq  to represent lines
in geometry.  Mathematica�s built-in  Line  graphics function serves some of the same
purposes, but can�t be used here.  It�s really intended for representing polygons, and isn�t
directly related to any linear equation.

2DGeometry specifies a line as usual, by a triple of linear-equation coefficients:
lineq[a,b,c]  is the line with equation  ax + by + c = 0.  A triple  a,b,c  consists of
valid line coefficients only when its first two entries are not both zero.  To draw figures,
clients will normally construct lines by using functions from this package that are
guaranteed to refer to valid triples.

The following definition helps the package react intelligibly to erroneous use of
expressions such as  lineq[0,0,c]:

lineq[0,0,c_] := noLine

Such errors result in Mathematica�s fruitlessly attempting to handle the undefined
symbol  noLine  as though it were a  lineq  object.  It usually outputs the offending
expression with  noLine  plainly visible.

Selectors are implemented for the coefficients of an equation  ax + by + c = 0
of a line  g:

c1[lineq[a_,b_,c_]] ^:= a
c2[lineq[a_,b_,c_]] ^:= b
c3[lineq[a_,b_,c_]] ^:= c

The equality and inequality operators for lines are implemented next.  Two
coefficient triples represent the same line if they�re dependent.  Here are the definitions:

lineq[a1_,b1_,c1_] == lineq[a2_,b2_,c2_] ^:=
  FullSimplify[a1*b2 == a2*b1 &&

     b1*c2 == b2*c1 && c1*a2 == c2*a1]

lineq[a1_,b1_,c1_] /= lineq[a2_,b2_,c2_] ^:=
  Not[lineq[a1, b1, c1] == lineq[a2, b2, c2]]

noLine == noLine ^= badEquation
noLine /=  noLine ^= badEquation

The following examples show these definitions in action.

in: g = lineq[1,2,3];  h = lineq[2,4,6];
j = lineq[2,4,7];  k = lineq[0,0,1];
{g == h, h == j, j /= j}
{j == k, k == k, k /= k}

out: {True, False, False}
{lineq[2,4,7] == noLine, badEquation, badEquation}

The second list of three examples consists of intentional errors.  The first output resulted
from the fact that  ==  has not been defined when one of its operands is the symbol
noLine.  You might reason that equation  k == k  would produce the analogous output
noLine == noLine.  But Mathematica automatically gives this the value  True.  That�s
why two special definitions were included to return the undefined symbol  badEquation.

Constructing  lineq  objects

The line through point  P = <x0, y0>  with slope  m  has equation

y � y0 = m (x � x0), i.e.,  m x � y + y0 � m x0 = 0.

Constructors are implemented for this and for the horizontal and vertical lines
through  P:        

pointSlope[P_,m_] := lineq[m,-1, c2[P] - m*c1[P]]

horizontal[P_] := pointSlope[P,0]

vertical[P_] := lineq[1,0,-c1[P]]

Examples:

in: P = Point[1,2];
{pointSlope[P, 1], horizontal[P], vertical[P]}

out: {lineq[1,-1,1], lineq[0,-1,2], lineq[1,0,-1]}

The outputs represent equations  x � y + 1 = 0,  � y + 2 = 0,  and  x � 1 = 0.

Stuctured similarly to  pointSlope  are two functions that represent the lines

h1: a1x + b1 y + c1 = 0 h2: a2x + b2 y + c2 = 0

through a point  P<x0, y0>  parallel and perpendicular to a given line

g: ax + b y + c = 0.

Since  h1 * g  and  h2 z g,  you can take  a1, b1 = a, b  and  a2, b2 = �b , a.  Since  P  lies
on  h1, h2,

0 = a1x0 + b1 y0 + c1 = ax0 + b y0 + c1,
0 = a2x0 + b2 y0 + c2 = �bx0 + a y0 + c2.

You can solve those equations for  c1, c2.  Here are the definitions of the corresponding
functions:



11.07  Analytic geometry with 2DGeometry MATHEMATICS & MATHEMATICA
Page 4 Chapter 11  2D graphics

1 You can enter Mathematica operator  w  by typing  \[Vee] .  It has no preassigned meaning.

2001-07-21  15:33

pointParallel[P_,g_] :=
  lineq[c1[g],c2[g], -c1[g]c1[P] - c2[g]c2[P]];

pointPerpendicular[P_,g_] :=
  lineq[-c2[g],c1[g], c2[g]c1[P] - c1[g]c2[P]];

You should provide examples of the behavior of these functions, as correctly used and
in error situations.

Join and intersection

The line  P w Q  through points  P = <x1, y1>  and   Q = <x2, y2>  that aren�t vertically
aligned has equation  y � y1 = m (x � x1),  where

m = .2 1

2 1

y y
x x
−
−

Algebraic manipulation converts that to the form

( y2 � y1) x  + (x1 � x2) y + (x2 y1 � x1 y2) = 0.   

This equation is correct even if  P  and  Q  are vertically aligned, as long as  P /= Q .  The
line  P w Q  is called the join of  P  and  Q .  2DGeometry implements this operation by
overloading the  w  operator1:

Point[x1_,y1_] w Point[x2_,y2_]
  ^:= lineq[y2-y1, x1-x2, x2*y1 - x1*y2]]

Examples:

in: P = Point[1,2];  Q = Point[3,4];
{P w Q, P w P}

out: {lineq[2,-2,2], noLine}

Nonparallel lines  g, h  with equations  a1x + b1y + c1 = 0,  a2x + b2y + c2 = 0
intersect when

Their intersection  g 1 h  consists of the single point  <x, y>.  The following definition
implements this concept via the  1  operator2:

lineq[a1_,b1_,c1_] 1 lineq[a2_,b2_,c2_] ^:= Module[{d},
  d = a1*b2-a2*b1;
  (1/d)Point[b1*c2-b2*c1, c1*a2-c2*a1]]

Examples (all but the first are intentional errors):

in: g = lineq[1,-1,0];  h = lineq[1,1,-1];
j = lineq[1,-1,1];  k = lineq[0,0, 1];
{g 1 h, g 1 g, g 1 j, g 1 k}

out3: {Point[1/2,1/2], noIntersection, noIntersection,
noLine 1 lineq[1,-1,0]}

Testing for concurrence

Reasoning in analytic geometry, such as the demonstration in the following section, often
involves deciding whether three lines are concurrent or mutually parallel.4  (For this
discussion, every line is regarded as parallel to itself.)  The question is whether a system

of equations for those lines has a solution  <x, y>  or if the lines are mutually parallel.
If there is a solution, then there�s a nontrivial solution  <xr,yr, zr>  of the related
equations

(take  x = xr,  y = yr,  and  z = 1),  hence

Conversely, if this determinant is zero, then there�s a nontrivial solution  <xr, yr, zr>
of equations (2).  If  zr /= 0  then  <x, y> = <xr/zr, yr/zr>  is a solution of equations
(1).  Finally, equations (2) have a nontrivial solution  <xr, yr, zr>  with  zr = 0  just in
case there�s a nonzero vector  <xr, yr>  perpendicular to all three lines.  Such a vector
exists just when the lines are mutually parallel.  In summary, determinant (3), built from
the coefficients of three lines, is zero if and only if the lines are concurrent or mutually
parallel.

This 2DGeometry function implements some of the considerations of the previous
paragraph:

concurrentQ[lineq[a1_,b1_,c1_],
 lineq[a2_,b2_,c2_],
 lineq[a3_,b3_,c3_]] ^:=

  FullSimplify[Det[{{a1,b1,c1},
   {a2,b2,c2},
   {a3,b3,c3}}] == 0]

It returns  True  just when the three lines are concurrent or mutually parallel.  Here are
some simple examples of its use:

in: g = lineq[1,2,3];  h = lineq[4,5,6];
P = g 1 h; j = pointPerpendicular[P,g];
k = pointParallel[origin,j];
l = pointParallel[2P,j];
m = lineq[0,0,1];
{concurrentQ[g,h,j], concurrentQ[g,h,k]}
{concurrentQ[j,k,l], concurrentQ[j,k,k]}
concurrentQ[k,l,m]

out: {True, False}
{True, True}
concurrentQ[lineq[-2,1,0], lineq[-2,1,8], noLine]

The input arranged for point  P  to lie on lines  g, h, j,  ensured that  g, h, k  are not
concurrent, and that  j, k, l  are mutually parallel.  The last two examples show that
coincident lines are regarded as parallel, and that  concurrentQ  reacts reasonably to
incorrect input.

In the previous examples all coordinates were integers.  What if some of them are
floating-point numbers?  For instance, change  g  to  lineq[0.9,2,3],  recompute  P  and
j,  and re-execute  concurrentQ[g,h,j].  You�ll get  False!  What has happened?  The
presence of floating-point input has forced Mathematica to use floating-point arithmetic
throughout, and round-off error has accumulated while computing determinant  d.  By
inserting a  Print  command in the  concurrentQ  code, you can see that  d .
�3.0×10�15,  which Mathematica evidently regards as nonzero.

You can usually trust  concurrentQ  to give correct answers when its input
coordinates are integers or algebraic formulas.  Even then, evaluation of this function
will fail if the determinant formula is so complicated that Mathematica can�t determine
whether its value is zero.  With floating-point input, you simply can�t trust
concurrentQ,  or other 2DGeometry functions that rely on testing whether some number
x  is zero.  An alternative approach would be for such functions to ask whether  x  is close
enough to zero.  But close enough is too hard to define to consider that in this book.  For
example, it would depend on the size of input values and probably on the intended use
of the output.

Additional operations

Section 11.2 defined function  distance  to compute the distance between two points.
Here, that name is overloaded to provide a function to compute the distance between a
point and a line5:

distance[Point[x_,y_], lineq[a_,b_,c_]] ^:= 
  Module[{d}, d = Sqrt[a^2 + b^2];
  Abs[a*x + b*y + c]/d]

Another function defined in the previous section returned the slope of the segment
determined by two  Point  objects.  Here, its name is overloaded to compute the slope
of a  lineq  object:

slope[lineq[a_,b_,c_]] ^:=
  If[FullSimplify[b == 0], noSlope, -a/b]

You can provide examples of the behavior of these functions, as correctly used and in
error situations.

The 2DGeometry package includes several more Boolean functions that test
relationships between points and lines:

horizontalQ   onQ    parallelQ    collinearQ
verticalQ    throughQ   perpendicularQ

Exercises 1, 3, and 5 specify their actions.6  You�re invited to try your hand at program-
ming them, and to compare your code with the package�s.

Exercises

1. Part 1.  2DGeometry includes functions that you can execute as follows, with any
lineq  and  Point  objects  g  and  P:

horizontalQ[g]   verticalQ[g]   onQ[P,g]   throughQ[g,P]

They�re all Boolean functions.  The first two return  True  just when  g  is horizon-
tal or vertical;  the second two, when it passes through  P.  Without consulting the
package, write code for these functions.  Test it on known  True  and  False  cases,
and on a case with invalid  g  input.  

Part 2.  Using floating-point input and various 2DGeometry functions,
construct lines  g, h  and point  Q = g 1 h  such that  onQ[Q,h]  is  False.  (This
shows that like  concurrentQ  the functions in this exercise are unreliable with
floating-point input.)

2. Write and test code for a Boolean function  between1Q  with  Point  arguments
P, Q, R.  It should return  True  just when  Q  lies between7  P  and  R.  Your code
should be based on the fact that this condition holds just when the sum of the
distances from  Q  to  P  and  R  is the distance between  P  and  R.  Test it on known
True  and  False  cases.  Find an example where it fails.  Why did it fail?  (If you
complete exercise 8, compare with that result.)

3. 2DGeometry includes functions that you can execute as follows, with any  lineq
objects  g, h:

perpendicularQ[g,h] parallelQ[g,h]

These Boolean functions return  True  just when  g  and  h  are parallel8 or perpen-
dicular.  Without consulting the package, write code for these functions.  Test it
on known  True  and  False  cases, and on a cases with invalid input.  Hint:  for
parallelQ  consider the code for the  ==  operation.  (Because these functions use
the  ==  operator, they�re not reliable with floating-point input.)

4. Part 1.  Construct a rhombus  OPQR,  where  O  is the origin,  as follows.  Select
two values of  θ  (in radians)  for which  Sin  and  Cos  return algebraic formulas;
they should not differ by any multiple of  π/2.  Use  cis  with these to construct
points  P  and  Q   on the unit circle.  Use 2DGeometry functions to construct  R.
Now use 2DGeometry functions to construct the diagonal lines of the rhombus,
and verify that they�re perpendicular.

 Part 2.  What happens if you choose rational  θ  values for which  Sin  and
Cos  don�t return formulas?  What if you use even one floating-point  θ  value?
Why?

5. 2DGeometry includes a function  collinearQ  such that  collinearQ[P,Q,R] is
True  just when all of the points  P, Q, R  fall on a single line.  Without consulting
the package, write code for it.  Test that with integer input on known  True  and
False  cases.  Find a case where it fails with floating-point input.  (Like other
2DGeometry functions that test for equality,  collinearQ  is unreliable with
floating-point input.)

6. Three points are collinear just when a certain determinant  d  involving their
coordinates is zero.  Function  collinearQ  computes  d.  Modify its code to output
d.  How can you predict the sign of  d  by looking at a picture of the points?

7. Write a Boolean function  betweenQ  that applies to three (real) numbers  t, u, v
and returns  True  just when  u  lies between  t  and  v.  Test it on known  True
and  False  cases.

8. Write and test code for a Boolean function  between2Q  with  Point  arguments
P, Q, R.  It should return  True  just when  Q  lies between  P  and  R .  Your code
should be based on the fact that this condition holds just when

i. they�re collinear, and

ii. one of these conditions holds:
(1) they�re horizontally aligned and their  x-coordinates fall in order, or
(2) they�re not, and their y-coordinates fall in order.

You may use the function  betweenQ  of exercise 7.  Test  between2Q  on known
True  and  False  cases.  Find an example where it fails.  Why did it fail?  (If you
completed exercise 2, compare with that result.)


